



# GESAMTSCHADENVERTEILUNG MIT DEM SAS-SYSTEM

KSFE BERLIN 21.2.197

---

## BEDEUTUNG DES GESAMTSCHADENS

---

Der Gesamtschaden als Summe aller Schäden für festes Kollektiv innerhalb eines festen Zeitraumes  
Verteilung wichtig für

- ⇒ Franchisen
- ⇒ Rückversicherung
- ⇒ Ruinwahrscheinlichkeiten
- ⇒ Reserveberechnungen

Seine Verteilung ist bedeutsam für die Gesamtbeurteilung des Portefeuilles und stellt daher eine wesentliche Aufgabe innerhalb der Risikotheorie dar.

---

## KOLLEKTIVES MODELL DER RISIKOTHEORIE

---

Dieses Modell läßt sich wie folgt beschreiben :

Zufallsvariable:


$N :=$  Schadenanzahl im Kollektiv in Periode mit Zähldichte  $(p_n)_{n \in \mathcal{N}}$

$X_i :=$  Höhe des  $i$ -ten Einzelschadens,  $i = 1, \dots, N$

mit den Annahmen :

$N, X_1, X_2, \dots$  sind stochastisch unabhängig

$X_i$  sind identisch verteilt mit Verteilungsfunktion  $F \quad \forall i \in \mathcal{N}_0$

gesucht: 

Gesamtschaden  $S^{koll} := \sum_{i=1}^N X_i$  mit Verteilungsfunktion  $G^{koll}$

Es betrachtet nur tatsächliche Schadenereignisse, ohne dabei auf die diesen Schäden zugrunde liegenden versicherungstechnischen Einheiten einzugehen. Insbesondere wird nicht betrachtet, welche Risiken keine Schäden, welche mehrere und wieviele Schäden verursachen.



---

EIGENSCHAFTEN

---

Die folgenden wichtigen Eigenschaften gelten für den Gesamtschaden und seine Verteilung:

$$\mathcal{Q}\left(\sum_{i=1}^N X_i \leq x\right) = G^{koll}(x) = \sum_{n=0}^{\infty} p_n \cdot F^{*n}(x)$$

$$\varphi_S(t) = g_N(\varphi_X(t)) \text{ mit}$$

$\varphi_X, \varphi_S$  momenterzeugende Funktionen  
 $g_N$  Verteilung erzeugende Funktion


$$E(S) = E(N) \cdot E(X)$$
$$V(S) = E(N) \cdot V(X) + V(N) \cdot (E(X))^2$$

Mit der Angabe als gewichteter unendlicher Summe von Faltungen sind die Berechnungsprobleme offenbar noch nicht gelöst. Die Relation zwischen den momenterzeugenden Funktionen kann zur Herleitung der Eigenschaften von Erwartungswert und Varianz genutzt werden und findet auch bei einigen verteilungsunabhängigen Verfahren zur Berechnung der Gesamtschadenverteilung Anwendung.



SYSTEMATISIERUNG DER BERECHNUNG

Die nachfolgende Grafik beschreibt die wesentlichen Verfahrensschritte und klassifiziert die derzeit gebräuchlichsten Verfahren :





---

PANJER-REKURSION

---

Wir betrachten im folgenden eine einfache Ausprägung des rekursiven Ansatzes:

Schlüssel ist die Beobachtung, daß die Zähldichten der gängigen Anzahlverteilungen einer bestimmten Rekursion genügen.

### Die Zähldichten der Anzahlverteilungen

○Poisson mit Parameter  $\lambda$   $a = 0$   $b = \lambda$

○Binomial mit Parametern  $n, p$   $a = -\frac{p}{1-p}$   $b = \frac{p}{1-p} \cdot (n + 1)$

○Negativ Binomial mit den Parametern  $r$  und  $p$   $a = p$   $b = p \cdot (r - 1)$

erfüllen eine Rekursion der Form

$$p_n = \left(a + \frac{b}{n}\right) \cdot p_{n-1} \quad (\text{Panjer})$$

(es gibt eine Vielzahl von Verallgemeinerungen)

Die Erweiterungen des Ansatzes betreffen die Ordnung der Rekursionsgleichung und verlangen die Gültigkeit erst ab einer festen Stelle an. Damit werden unter Erhalt der rekursiven Berechnungsmöglichkeit komplexere Verteilungen an die Schadenanzahl anpaßbar. (z. B. Delaport-Verteilung)



---

REKURSIONSFORMEL

---

Mit Hilfe der Panjer-Rekursion und des Zusammenhangs der momenterzeugenden Funktionen läßt sich eine Rekursion für die Dichte des Gesamtschadens herleiten.

$N$  sei  $\mathcal{R}_0$ -Zufallsvariable mit Zähldichte  $(p_n)_{n \in \mathcal{R}_0}$ ,  
die die Panjer - Rekursion erfüllt.

$X_1, X_2, \dots$  stochastisch unabhängig, identisch verteilt

$\Delta$  - arithmetische Zufallsvariable mit

Zähldichte  $(f_i)_{i \in \mathcal{R}_0} = (\mathcal{Q}(X = i \cdot \Delta))_{i \in \mathcal{R}_0}$  und Träger  $\{x_i = i \cdot \Delta : i \in \mathcal{R}_0\}$ .

Dann gilt für :  $S = \sum_{i=1}^N X_i$  mit Zähldichte  $(g_n)_{n \in \mathcal{R}_0}$ :

$$\mathcal{Q}(S = n \cdot \Delta) = g_n = \frac{1}{1 - a \cdot f_0} \cdot \sum_{j=1}^n \left[ \left( a + \frac{b \cdot j}{n} \right) \cdot f_j \cdot g_{n-j} \right]$$

mit dem Rekursionsanfang:  $g_0 = f_0$

Dies ist die Basis für den nachfolgenden Algorithmus .



---

DAS SAS-PROGRAMM (MIT IML)

---

Die Umsetzung der dargestellten Rekursion erfolgt innerhalb des SAS-Systems mit der interaktiven Matrixsprache, die ein hohes Maß an Flexibilität und Mächtigkeit aufweist.

Nachfolgend das komplette kommentierte Programm am Beispiel einer künstlich erzeugten Schadenhöhenverteilung.

```
/* Bereitstellung einer theoretischen Einzelschadenhöhendatei mit Mittel 100 */  
/* im Praxisprozess nach Modellierung der Einzelschadenhöhen */
```

```
data schh ;  
  do j=1 to 1000;  
  
    k=poisson(200,j);  
    i=j-100;  
    w=k-lag(k);  
    output;  
  end;  
run;
```

```
/* zu kleine Wahrscheinlichkeiten entfernt */
```

```
data schh (keep= i w);  
set schh;  
if i>0;  
run;
```

```
/* Ende der Bereitstellung einer Einzelschadenhöhendatei */
```

```
proc iml;
```

```
/* Einlesen der SAS-Test-Datei in IML*/
```

```
close schh;  
/* vorsichtshalber schließen, um die richtige Zeigerposition zu haben */
```

```
use schh;  
read current var {w} into snull;  
/* Wahrscheinlichkeit für Schäden der Höhe Null gesondert einlesen */
```

```
read after var {w} into s;  
/* Rest in den Vektor s einlesen */
```

```
/* Ergebnisvektor initialisieren */  
f=0;
```

```
/* erstes und zweites Moment der Schadenhöhenverteilung berechnen */
```

```
m1= (1:nrow(s))* s ;  
m2= ((1:nrow(s))#(1:nrow(s))) *s ;
```

```
/* Type der Schadenanzahlverteilung festlegen */  
/* zur Auswahl stehen poisson und negbin */
```



GESAMTSCHADENVERTEILUNG MIT DEM SAS-SYSTEM

12.01.2006

```
typ="negbin";

/* Paramter für den Typ der Schadenanzahlverteilung setzen */
/* Mittelwert und Varianz der Schadenanzahl */

manz=100;
vanz=250;

/* Poisson-Fall */
if typ="poisson" then do;
    a=0;
    b=manz;
    f[1]=exp(b*snull-b);
    /* Ergebnisvektor f initialisieren */
    /* f[1]= Prob [gesaufw =0]          */
    /* entsteht durch Diskretisierung */
    mges=b*m1;
    vges=b*m2;

    end;

/* negbin - Fall */
if typ="negbin" then do;
    /* falls die Parameter nach der Momenten-Methode */
    /* berechnet werden sollen, sonst direkt setzen */
    a=(vanz-manz)/vanz;
    b=(manz*manz+manz-vanz) /vanz;
    f[1]=(1+ (a*((a+b)/(1-a))*(1-snull)))/(a+b)##((-a-b)/a);
    /* Ergebnisvektor f initialisieren */
    /* f[1]= Prob [gesaufw =0]          */
    /* entsteht durch Diskretisierung */
    mges=(a+b)/(1-a)*m1;
    vges=((a+b)/(1-a))*m2+m1##2*(a*(a+b))/(1-a)##2;

    end;

/* jetzt läßt sich etwa die 3 sigma -Regel anwenden */

/* drei-sigma-Regel */
do j=1 to ceil(mges+3*sqrt(vges));

/* Anzahl der Werte hier sinnvoll festzulegen */

k=min(j,nrow(s));

/* stets nur solange wie Einzelschadenhäufigkeiten vorhanden */

p=T(1:k) * b/j+a || s[1:k] || f[1:k];
/* eine dreispaltige Matrix wird aufgebaut */
/* sie enthält die Gewichtungsfaktoren, die Einzelschadenhöhen und */
/* die bisher berechneten Gesamtschadenhöhen */

x=p[,#][+,]/(1-a*snull);
/* die Summe über die Produkte wird mit den      "subscript reduction operators" gebildet */
```



GESAMTSCHADENVERTEILUNG MIT DEM SAS-SYSTEM

12.01.2006

```
f=x//f;
/* der kleine Trick ! */
/* f wird in Kellerstruktur abgelegt */
/* es ist so keine komplexen Indizes zu führen */
/* alles steht richtig bereit für den nächsten Schritt ! */

end;

/* Abspeichern des Ergebnisses in der Datei work.erges */

create erges from f;
append from f;

quit ;

/* IML verlassen */

/* die Reihenfolge ist wegen der Kellerstruktur nun umzukehren */

/* Index k einführen */
data erges;
set erges;
k=_N_;
run;

/* absteigend nach k sortieren */
proc sort data=erges; by descending k;run;

/* Ergebnisdatei indiziert ablegen, überflüssigen Index löschen */
data erges (drop=k);
set erges (rename=(col1=wges));
i=_N_ -1;

run;

/* indiziert mit Index i hat man nun */
/* wges= Prob( Gesamtsschaden= i ) */

/* grafische Darstellung des Ergebnisses */
symbol1 interpol=join value=dot height=0.3;
proc gplot data=erges;
plot wges*i;
run;
quit;
```





---

ZUSAMMENFASSUNG

---

Für Gesamtschadenverteilung haben Sie erhalten :

- ✓ Einführung in Problemstellung
- ✓ eine elegante und schnelle theoretische Lösung
- ✓ praktische Umsetzung in **SAS** mit gutem Laufzeitverhalten

Bei dem Gesamtprozess werden in den ersten beiden Schritten Anpassungsfehler und Diskretisierungsfehler auftreten. Dann ist die Berechnung exakt.

Bei mehreren Teilportefeuilles sind diese durch Faltung zusammenzufassen.

Hier haben wir gute Erfahrungen bei der Nutzung der schnellen Fouriertransformationen in SAS gemacht.

Literatur :

Klaus Jürgen Schröter :

Verfahren zur Approximation der Gesamtschadenverteilung

VVW Karlsruhe 1995